
     A                                      DSPSIZ(24 80 *DS3)                  

     A                                      INDARA                              

     A                                      PRINT                               

     A          R SFL                       SFL                                 

     A            WTHRDATE      10A  O 15  2                                    

     A            DESC          20A  O 15 15                                    

     A            TEMPLO         3A  O 15 39                                    

     A            TEMPHI         3A  O 15 46                                    

     A            PRECPMO        3A  O 15 54                                    

     A            PRECPNT        3A  O 15 61                                    

     A          R SFLCTL                    SFLCTL(SFL)                         

     A                                      CF01(01)                            

     A                                      CF12(12)                            

     A                                      OVERLAY                             

     A  62                                  SFLDSP                              

     A  62                                  SFLDSPCTL                           

     A  60                                  SFLCLR                              

     A                                      SFLSIZ(0008)                        

     A                                      SFLPAG(0007)                        

     A                                  2  2DATE                                

     A                                      EDTCDE(Y)                           

     A                                      COLOR(BLU)                          

     A                                  2 20'www.cdyne.com SOAP Weather Forecas-

     A                                      t Inquiry'                          

     A                                      DSPATR(RI)                          

     A                                      COLOR(BLU)                          

     A                                  2 72TIME                                

     A                                      COLOR(BLU)                          

     A                                  4 13'Zip Code ...'                      

     A            ZIPCODE        5S 0O  4 27                                    

     A                                  6 13'City .......'                      

     A            CITY          20A  O  6 27                                    

     A                                  7 52'Press F1 for hourly forecast'      

     A                                  8 13'State ......'                      

     A            STATE          2A  O  8 27                                    

     A                                  8 52'Press F12 to Return'               

     A                                 10 13'Weather Station City ...'          

     A            WTHRSTA       20A  O 10 38                                    

     A                                 13 39'Temperature'                       

     A                                 13 52'Prob of Precip'                    

     A                                 14  2'Date         Description          -

     A                                         Low   High    Morn   Night'      

     A                                      DSPATR(UL)                          

     A          R SCREEN1                                                       

     A                                      CF03(03)                            

     A                                  2  2DATE                                

     A                                      EDTCDE(Y)                           

     A                                      COLOR(BLU)                          

     A                                  2 20'www.cydne.com SOAP Weather Forecas-

     A                                      t Inquiry'                          

     A                                      DSPATR(RI)                          

     A                                      COLOR(BLU)                          

     A                                  2 72TIME                                

     A                                      COLOR(BLU)                          

     A                                  7 21'Enter Zip Code .....'              

     A            ZIPCODE        5N 0B  7 42                                    

     A  71                             10 21'Zip Code cannot be zero, please re-

     A                                      -enter!'                            

     A  73                             11  7'An error occurred when calling CDY-

     A                                      NE service, please try again later.'

     A  72                             12 23'Zip Code not found, please re-ente-

     A                                      r!'                                 

     A                                 14 19'Press F3 to End Job'               


